U.S. Department of Agriculture
Forest Service General Technical Report SO-9

Combining-Ability Determinations for Incomplete Mating Designs

E. B. Snyder

Southern Forest Experiment Station
Forest Service
U.S. Department of Agriculture

1975

COMBINING-ABILITY DETERMINATIONS FOR INCOMPLETE MATING DESIGNS

E. B. Snyder[1]

It is shown how general combining ability values (GCA's) from cross-, open-, and self-pollinated progeny can be derived in a single analysis. Breeding values are employed to facilitate explaining genetic models of the expected family means and the derivation of the GCA's. A FORTRAN computer program also includes computation of specific combining ability values and several options.

Additional keywords: Diallel cross, reciprocal cross, specific combining ability, *panmixia*.

The plant breeder ranks parents according to their general combining abilities. The calculation of mean effects is complicated when all parents of a particular set have not been crossed systematically, but theories and procedures for analyzing even severely unbalanced data are nevertheless available (Bohren et al. 1965, Gilbert 1967).

The mathematics of such theories may present difficulties to the practical breeder. Here, I advance a simplified explanation by showing how breeding values (2 GCA + $\overline{X}$ terms) can be manipulated for analyzing data from cross, self, and open pollinations in a single analysis. I will also discuss the usefulness of selfing, a means of testing the randomness of open-pollination, a technique for estimating the population mean, and a method for combining data from several experiments.

As computer programs for such unbalanced data have generally not been published, a versatile, efficient FORTRAN program is described along with substitute procedures requiring no more than a desk calculator.

[1]The author is Principal Plant Geneticist, Southern Forest Experiment Station, Forest Service—USDA, Gulfport, Mississippi. He is indebted to W. L. Nance, Southern Forest Experiment Station, for much guidance, and to N. E. Gilbert, University of British Columbia, and N. G. Alvey, Rothamsted Experiment Station, England, for test data and comments.

THEORY AND APPLICATIONS

When reciprocal crosses are pooled, a model for determining general combining ability effects (GCA's) from intercrossed parents is:

$$y_{ijk} = \mu + gca_i + gca_j + \varepsilon_{ijk}$$

where y_{ijk} = k^{th} individual in the i, j^{th} cross
gca_i = GCA effect of the i^{th} parent
gca_j = GCA effect of the j^{th} parent

with the assumptions that

$$\Sigma n_i gca_i = 0$$
$$\{\varepsilon_{ijk}\} \sim NID\ (0, \sigma^2)$$

where n_i = the number of plants per family (given a value of 1 in unweighted analyses).

The GCA estimates are applicable only to the restricted set of parents tested, i.e., the model is fixed.

Breeding values (2 GCA+$\overline{X}$) are easily visualized and hence facilitate explaining the genetic model. A value of the i^{th} parent is (2 GCA_i+$\overline{X}$), or twice the parental contribution to an individual progeny. The models for the expected family means (mid-parental values) of individual progeny of various types are:

Cross-pollinated $$\frac{(2\,GCA_i+\overline{X})+(2\,GCA_j+\overline{X})}{2} = GCA_i+GCA_j+\overline{X}$$

Self-pollinated $$\frac{(2\,GCA_i+\overline{X})+(2\,GCA_i+\overline{X})}{2} = 2\,GCA_i+\overline{X}$$

Open-pollinated $$\frac{(2\,GCA_i+\overline{X})+(2\,GCA_{op}+\overline{X})}{2} = (GCA_i+GCA_{op}+\overline{X}$$

or (GCA_i+$\overline{X}$), because GCA_{op}=0.

Since formulae for all three family types are built up from breeding values, observed means can be used to calculate breeding values for each family type. Furthermore, breeding values for each type can be entered independently or together in a single diallel analysis, thereby giving more replication and confidence to estimates.

Selfs should be incorporated only if there is no inbreeding depression. As the structural model shows, an extreme breeding value is doubled among selfed progeny, whereas among crosses it is diluted by being paired with a less extreme value. Thus, the extremeness is a "selfing" effect only in the sense that a single extreme breeding value (variant) would not be so conspicuous in cross-pollinated progeny. If there were no inbreeding effect or if it could be compensated for, selfing would efficiently and unambiguously identify extreme parents.

Provided there are some parents in common, populations of various types of material from different experiments can be used in a single diallel table by adjusting breeding values and, from these, the phenotypes to be integrated. Combining populations thus requires the preliminary analysis of each and subsequently calculating the adjusting ratios from the average of breeding values in common. Similarly, data from separate plantings of the same families may be combined after appropriate adjustments.

Data from open-pollinated (OP) families can be included in the same analysis with those from cross-pollinated families. This is possible when wind-borne pollen is treated as that from a single male parent representing the population. Two values of interest to the breeder can be deduced from the OP model. Deviation of the GCA_{op} value from zero measures the deviation of selected females from randomness. Also, the mean of the population is estimated by the OP breeding value.

The computer calculations are based on the usual least-squares methods suggested by Gilbert (1967) and illustrated by England (1974). If a computer is not available, iteration with a calculating machine will serve. Thus, for unweighted analysis with reciprocals absent or pooled and no selfs (Yates 1947), parental mean of i^{th} parent, $\overline{P}_i \cong$

$$\frac{\text{Breeding value of } i^{th} \text{ parent} + \overline{BV} \text{ of other parents crossed with } P_i}{2}$$

For example, the breeding value for parent 3 of a 13-parent half-diallel with no selfs becomes:

$$(2\ GCA_3+\overline{X})=2\overline{P}_3-1/12(2GCA_1+\overline{X}+2\ GCA_2+\overline{X}+2GCA_4+\overline{X}\ldots+2GCA_{13}+\overline{X})$$

The $\overline{P}_3$ value is the mean of all families for which P_3 is a parent. When the initial equations are set up, $\overline{P}_i$ values must be substituted for the $(2GCA_i+\overline{X})$ values within the righthand parentheses. As the analysis continues, $(2GCA_i+\overline{X})$ values are entered as soon as determined. Usually, satisfactory convergence will be achieved with five or fewer iterations.

Once the breeding values are found, specific combining abilities may be computed directly in the usual way:

$$SCA=\text{observed mean of } i^{th},\ j^{th} \text{ family} - \frac{(2\ GCA_i+\overline{X})+(2GCA_j+\overline{X})}{2}$$

By definition, SCA's are not appropriate for selfs.

Because the mid-parental expectation of a cross is the last term in the formula, definition and use of $(2\ GCA+\overline{X})$ breeding values is easy. They are simply parental values which, when averaged, predict performance of the cross. The convenience and ease of visualization for instruction and breeding are lacking with GCA or 2 GCA deviations alone.

THE PROGRAM

The mathematics of the program is similar to that used by most statisticians—least-squares equations are formed and GCA values are solved for by inverting the matrix according to the elimination method. For some limitations to the program, Gilbert (1967) should be consulted. Solutions may differ slightly depending on the degrees of design balance and order in which data are fitted. Also, if it is desired to estimate values separately within male or female sets, Milliken et al. (1970) should be consulted.

The usefulness of the program lies with its supplementary options and listings:

Sum or mean data with or without the number of plants per family may be entered for as many as 50 parents. Options are offered for weighting by number of plants, not weighting, or both. If no reciprocal crosses are present, data are entered as upper-triangle matrix elements. Reciprocal data are entered in the lower triangle and will automatically be accumulated or averaged into the upper triangle.

Input or data errors will be signalled, after which the analysis will be terminated and the next one started. Warnings of some types of singular or near-singular solutions are given. The program is generously supplied with comment cards that should facilitate modification by others.

Listed in addition to combining ability and breeding values are: the data entered, family means, number of parents, effective number of families per parent, mean breeding value, and experimental mean.

The inverse is printed for possible use in obtaining confidence intervals for the GCA values (Milliken et al. 1970). An error mean square must be available from a separate variance-analyzing program. The program of Schaffer and Usanis (1969) is adequate and so versatile that we do not supply variance terms except for a GCA sum of squares for comparison with it and other programs.

LITERATURE CITED

Bohren, B. B., H. E. McKean, and G. W. Friars.
1965. The expected mean squares in genetic experiments when only one parent is identified. Biometrics 21: 436-446.

England, F.
1974. A general approximate method of fitting additive and specific combining abilities to the diallel cross with unequal numbers of observations in the cells. Theor. and Appl. Genet. 44: 378-380.

Gilbert, N. E.
1967. Additive combining abilities fitted to plant breeding data. Biometrics 23: 45-49.

Milliken, G. A., H. L. Bush, A. W. Ericksen, and A. Suzaki.
1970. Estimating general combining ability from an incomplete crossing system. J. Am. Soc. Sugar Beet Technol. 16: 264-274.

Schaffer, H. E., and R. A. Usanis.
1969. General least squares analysis of diallel experiments, a computer program—DIALL. N.C. State Univ. Genet. Dep. Res. Rep. 1, 61 p.

Yates, F.
1947. Analysis of data from all possible reciprocal crosses between a set of parental lines. Heredity 1: 287-301.

```
C     E.B.SNYDER-W.NANCE.SFES THIS PROGRAM IS FOR ANY TYPE OF DIALLEL
C     BUT ASSUMES NO SELFING EFFECTS.SUMS WITH PLANT COUNTS/FAMILY,AND
C     MEANS WITH OR WITHOUT COUNTS MAY BE ENTERED TO OBTAIN WEIGHTED OR
C     UNWEIGHTED VALUES.ALL TYPES OF DATA ARE ACCUMULATED IN UPPER TRI-
C     ANGLE OF MATRICES FOR ANALYSIS
C
C     **************
C     * CARD ORDER *
C     **************
C
C        CONTROL CARD        **
C        INPUT FORMAT CARD     * REPEATED FOR EACH EXPERIMENT
C        TITLE CARD            *
C        DATA CARDS          **
C
C        TRAILER CARD ** CONSISTS OF A BLANK CARD USED ONLY AFTER FINAL
C                                      DATA SET
C
C     ****************
C     * CONTROL CARD *
C     ****************
C
C       COLS.1-7 CONTROL
C       COLS.8-10 PROBLEM NO.
C       COLS.11-13 NUMBER OF PARENTS
C       COLS.14-17 NO. OF FAMILIES WITH DATA--LEAVE BLANK IF COL.19 IS 0
C
C                         * 1 IF ID IS PROVIDED WITH DATA
C       COLS.18-19 PUNCH*
C                         * 0 IF NO ID IS PROVIDED. IN THIS CASE ZEROS
C                             MUST BE ENTERED WHERE MISSING DATA
C
C                          *1 IF DATA ENTERED ARE SUMS AND PLANTS/FAMILY
C       COL.    21 PUNCH * 2 IF DATA ENTERED ARE MEANS AND PLANTS/FAMILY
C                          *3 IF DATA ENTERED ARE MEANS ONLY
C
C                          *1 IF ANALYSIS WEIGHTED BY PLANTS/FAMILY
C       COL.    23 PUNCH * 2 IF WEIGHTED AND UNWEIGHTED ANALYSIS
C                          *3 IF UNWEIGHTED ANALYSIS
C
C     *********************
C     * INPUT FORMAT CARD *
C     *********************
C
C        PLANTS/FAMILY,IF USED,ARE ENTERED IN F FORMAT
C
C     **************
C     * TITLE CARD *
C     **************
C
C       COLS.1-5 TITLE
C       COLS.8-72 TEXTUAL IDENTIFICATION OF THE VARIABLE OR CHARACTER
C
C     **************
C     * DATA CARDS *
C     **************
C
C       ONE TRAIT AT A TIME IS ANALYZED. MEASUREMENT DATUM FOR A FAMILY
C       IS PREFERABLY PRECEDED BY PARENTAL COMPOSITION ID,WHERE FEMALE
C       (CODED SEQUENTIALLY 1 TO N) IS IMMEDIATELY FOLLOWED BY A SIMIL-
C       ARLY CODED MALE,E.G.0102 GIVES BOTH PARENTAGE AND MATRIX INDI-
C       CES. THE MEASUREMENT DATA MAY BE FOLLOWED IN THE SAME FIELD BY
C       NUMBER OF PLANTS/FAMILY. SEE OPTIONS LISTED FOR CONTROL CARDS
C       ABOVE.THE DATA FROM MORE THAN ONE PLANT CAN BE PLACED ON ONE
```

```
C          CARD ONLY IF ID'S ARE NOT SUPPLIED BUT HERE THERE ARE NO INPUT
C          ERROR CHECKS. ALL DATA ARE ENTERED BY ROWS AND, IF THERE ARE NO
C          RECIPROCAL CROSSES,IN THE UPPER TRIANGLE. WHERE THE TRUE VALUE
C          OF DATUM IS ZERO, A VERY SMALL NUMBER MUST BE ENTERED INSTEAD.
C
C***********************************************************************
C
C
      REAL NN,MM,KK
      INTEGER P1,DE,WT,W,EMPTY,CONTRO
      DIMENSION U(50,50), X(50,50), IX(50,50), KK(50,50), MM(50,50), NN(
     150,50), Q(50,50), Y(50,1), N(50), P(50), B(50), G(50), FMT(12), TI
     2TLE(11), XPY(50,1)
C     FOR SOME COMPUTERS THE FOLLOWING MAY BE NEEDED
C     DIMENSION RAY(50,50), BB(50,1), JJ(50)
      EQUIVALENCE (IX,X)
      DATA CONTRO /6HCONTRO/
      DATA IOU,IIU /6,5/
      DATA EMPTY /5HEMPTY/
      WRITE (IOU,530)
    5 READ (IIU,570,END=520) IA
      IF (IA.NE.CONTRO) GO TO 60
   10 READ (0,540) P1,NF,ID,DE,WT
   20 IF (P1.EQ.0) GO TO 520
      READ (IIU,550) (FMT(I1),I1=1,12)
      READ (IIU,590) (TITLE(I3),I3=1,11)
      DO 30 IJK=1,P1
        DO 30 IKL=1,P1
          NN(IJK,IKL)=0.
          IX(IJK,IKL)=EMPTY
   30     CONTINUE
      IF (ID.EQ.1) GO TO 40
      IF (ID.EQ.0.AND.DE.LT.3) READ (IIU,FMT) ((X(I,J),NN(I,J),J=1,P1),I
     1=1,P1)
      IF (ID.EQ.0.AND.DE.EQ.3) READ (IIU,FMT) ((X(I,J),J=1,P1),I=1,P1)
      GO TO 80
   40 DO 50 IJK=1,NF
        IF (DE.LT.3) READ (IIU,FMT) IFEM,IMAL,XIN,XNNIN
        IF (DE.EQ.3) READ (IIU,FMT) IFEM,IMAL,XIN
        IF (IFEM.LE.0.OR.IFEM.GT.P1) GO TO 60
        IF (IMAL.LE.0.OR.IMAL.GT.P1) GO TO 60
        IF (IX(IFEM,IMAL).NE.EMPTY) GO TO 60
        X(IFEM,IMAL)=XIN
        IF (DE.LT.3) NN(IFEM,IMAL)=XNNIN
   50   CONTINUE
      GO TO 80
   60 WRITE (IOU,560)
C     SEARCH FOR OTHER PROBLEMS SUBMITTED BUT AN ERROR CAN RUIN THE
C     READING IN OF IMMEDIATELY FOLLOWING PROBLEM(S)
   70 READ (IIU,570,END=520) IA
      IF (IA.NE.CONTRO) GO TO 70
      READ (0,540) P1,NF,ID,DE,WT
      GO TO 20
   80 WRITE (IOU,580)
      WRITE (IOU,590) (TITLE(I3),I3=1,11)
      WRITE (IOU,600) P1
      DO 90 I1=1,P1
        DO 90 I2=1,P1
          IF (IX(I1,I2).EQ.EMPTY) X(I1,I2)=0.
   90     CONTINUE
      KOUNT=0
      DO 100 I=1,P1
        DO 100 J=1,P1
          U(I,J)=X(I,J)
  100     CONTINUE
```

```
          IF (DE.NE.1) GO TO 120
          DO 110 I=1,P1
            DO 110 J=1,P1
              IF (NN(I,J).EQ.0.) GO TO 110
              X(I,J)=X(I,J)/NN(I,J)
  110         CONTINUE
C         TEST FOR RECIPROCALS IN LOWER TRIANGLE
  120     DO 130 I=2,P1
            L=I-1
            DO 130 J=1,L
  130         IF (X(I,J).GT.0.) GO TO 420
          W=0.
C         GIVES NATURE OF DATA ENTERED AND WEIGHTING DESIRED
  140     IF (DE.NE.1) GO TO 150
          WRITE (IOU,610)
  150     IF (DE.NE.3) GO TO 190
          WRITE (IOU,620)
          IF (DE.EQ.3.AND.WT.EQ.3) GO TO 160
          WRITE (IOU,630)
  160     WRITE (IOU,640)
C         ENTER 1'S IN NN MATRIX IF UNWEIGHTED ANALYSIS
          DO 170 IJK=1,P1
            DO 170 IKL=1,P1
  170         NN(IJK,IKL)=0.
          DO 180 I=1,P1
            DO 180 J=1,P1
              IF (X(I,J).NE.0.) NN(I,J)=1.
  180         CONTINUE
          GO TO 210
  190     IF (DE.EQ.1) GO TO 200
          WRITE (IOU,650)
  200     IF (WT.EQ.3) GO TO 160
          WRITE (IOU,660)
  210     DO 220 I=1,P1
            DO 220 J=I,P1
  220         KK(I,J)=NN(I,J)
C         ADJUST FREQ OF SELFS IN DIAG OF NN
          DO 230 I=1,P1
            NN(I,I)=4.*NN(I,I)
  230       CONTINUE
          KOUNT=KOUNT+1
          DO 240 I=1,P1
            N(I)=0
            P(I)=0.
  240       B(I)=0.
          N2=0
C         START MAIN PROGRAM BY SETTING SELF DATA IN ACCUMULATORS
          DO 250 I=1,P1
            N(I)=0.5*NN(I,I)
            P(I)=0.5*NN(I,I)*X(I,I)
  250       CONTINUE
C         ACCUMULATE YIELDS AND COUNTS FOR EACH PARENT
          DO 260 J=2,P1
            L=J-1
            DO 260 I=1,L
              IF (X(I,J).EQ.0.) GO TO 260
              N(I)=N(I)+1
              N(J)=N(J)+1
              P(I)=P(I)+NN(I,J)*X(I,J)
              P(J)=P(J)+NN(I,J)*X(I,J)
  260         CONTINUE
          DO 270 I=1,P1
            N2=N2+N(I)
  270       CONTINUE
C         ARRANGE P(J) IN MATRIX AND PRINT
```

```
      DO 280 J=1,P1
        Y(J,1)=2.*P(J)
  280   CONTINUE
      DO 290 I=1,P1
        XPY(I,1)=Y(I,1)
  290   CONTINUE
C     DUPLICATE UPPER NN ELEMENTS IN LOWER TRIANGLE
      DO 300 I=1,L
        DO 300 J=2,P1
  300     NN(J,I)=NN(I,J)
C     ACCUMULATE TOTAL ADJUSTED COUNTS IN DIAGONALS
      DO 320 J=1,P1
        N(J)=0.
        DO 310 I=1,P1
          N(J)=N(J)+NN(J,I)
  310     CONTINUE
        NN(J,J)=N(J)
  320   CONTINUE
      WRITE (IOU,670)
      WRITE (IOU,680) (NN(I,I),I=1,P1)
      CALL MATINV (NN,P1,Y,1,DETERM,NDEP,1)
      WRITE (IOU,690) NDEP
      S=0.
      DO 330 I=1,P1
        B(I)=Y(I,1)
        S=S+B(I)
  330   CONTINUE
      XP1=P1-NDEP
      S=S/XP1
      IF (NDEP.EQ.0) GO TO 335
      S=S/2.0
      JXP1 = XP1
      DO 332 I = 1,JXP1
      B(I)=B(I)-S
  332   CONTINUE
  335 WRITE (IOU,700) S
C     EAN AND SS TERMS LINK WITH ANV
      SUM1=0.
      SUM2=0.
      SUM3=0.
      DO 340 I=1,P1
        DO 340 J=1,P1
          SUM1=SUM1+X(I,J)*KK(I,J)
          SUM2=SUM2+KK(I,J)
  340     CONTINUE
      DO 350 I=1,P1
        SUM3=SUM3+XPY(I,1)*Y(I,1)
  350   CONTINUE
        M1=SUM1/SUM2
      SUM2=((SUM1)**2)*SUM2
      SUM3=(SUM3/4.0)-SUM2
      WRITE (IOU,710)
      WRITE (IOU,720) SUM2
      WRITE (IOU,730) SUM3
      WRITE (IOU,740) SUM1
      DO 360 J=1,P1
      IF (B(J).EQ.0.) B(J)=S
        G(J)=((B(J)-S)/2.0)
  360   CONTINUE
      DO 370 I=1,P1
        L=I+1
        DO 370 J=L,P1
          IF (X(I,J).EQ.0.) Q(I,J)=0.
          IF (X(I,J).NE.0.) Q(I,J)=X(I,J)-(B(I)+B(J))/2.0
  370     CONTINUE
```

```
      WRITE (IOU,750)
      CALL MPRINT (NN,P1,1,6HNN MAT)
      IF (DE.GT.1.AND.W.EQ.0) GO TO 390
      IF (DE.EQ.1.AND.W.EQ.0) GO TO 380
      WRITE (IOU,760)
      CALL MPRINT (MM,P1,1,6HMM MAT)
      WRITE (IOU,770)
      CALL MPRINT (U,P1,1,6H U MAT)
      GO TO 390
  380 WRITE (IOU,780)
      CALL MPRINT (U,P1,1,6H U MAT)
  390 WRITE (IOU,790)
      WRITE (IOU,800) (G(I),I=1,P1)
      WRITE (IOU,810)
      WRITE (IOU,820) (B(I),I=1,P1)
      WRITE (IOU,830)
      LINES=0
      DO 410 I=1,P1
        DO 410 J=I,P1
          IF (KK(I,J).EQ.0.) GO TO 410
          LINES=LINES+1
          IF (LINES.NE.51) GO TO 400
          LINES=0
          WRITE (IOU,830)
  400     WRITE (IOU,840) I,J,KK(I,J),X(I,J),Q(I,J)
  410     CONTINUE
      IF (WT.EQ.2.AND.KOUNT.LT.2) GO TO 160
      GO TO 5
C     THIS ENDS ANALYSIS
  420 W=1.
C     THE FOLLOWING PUTS RECIPS IN HALF DIALLEL FORM
      IF (DE.LT.3) GO TO 450
      DO 430 IJK=1,P1
        DO 430 IKL=1,P1
  430     NN(IJK,IKL)=0
      DO 440 I=1,P1
        DO 440 J=1,P1
  440     IF (X(I,J).NE.0.) NN(I,J)=1.
  450 DO 460 I=1,P1
        DO 460 J=1,P1
          X(I,J)=NN(I,J)*X(I,J)
  460     CONTINUE
      DO 470 I=1,P1
        DO 470 J=1,P1
  470     MM(I,J)=NN(I,J)
      DO 480 I=1,P1
        DO 480 J=1,P1
          NN(I,J)=NN(I,J)+NN(J,I)
          X(I,J)=X(I,J)+X(J,I)
  480     CONTINUE
      DO 490 I=1,P1
        NN(I,I)=0.5*NN(I,I)
        X(I,I)=0.5*X(I,I)
  490   CONTINUE
      DO 500 I=2,P1
        L=I-1
        DO 500 J=1,L
          NN(I,J)=0.
          X(I,J)=0.
  500     CONTINUE
      DO 510 I=1,P1
        DO 510 J=I,P1
          IF (X(I,J).EQ.0.) GO TO 510
          X(I,J)=X(I,J)/NN(I,J)
```

```
      510     CONTINUE
        GO TO 140
      520 STOP
C
      530 FORMAT (1H1,61H        GCA AND SCA VALUES FOR COMPLETE OR INCOMPLET
     1E DIALLELS)
      540 FORMAT (10X,I3,I4,3I2)
      550 FORMAT (12A6)
      560 FORMAT (1H0,65H END OF ANALYSIS OR TERMINATION DUE TO INPUT FORMAT
     1 OR DATA ERROR)
      570 FORMAT (A6)
      580 FORMAT (1H0,51H * * * * * * * * * * * * * * * * * * * * * * * * *
     1)
      590 FORMAT (1H0,6X,11A6)
      600 FORMAT (1H0,25H THE NUMBER OF PARENTS IS,I4)
      610 FORMAT(1H0,41H DATA ENTERED WERE SUMS AND PLANTS/FAMILY)
      620 FORMAT (1H0,29H DATA ENTERED WERE MEANS ONLY)
      630 FORMAT (1H0,31H WEIGHTED ANALYSIS NOT POSSIBLE)
      640 FORMAT (1H0,20H UNWEIGHTED ANALYSIS)
      650 FORMAT (1H0,46H DATA ENTERED WERE MEANS AND PLANTS PER FAMILY)
      660 FORMAT (1H0,18H WEIGHTED ANALYSIS)
      670 FORMAT (1H0,82H THE EFFECTIVE NUMBER OF FAMILIES OR PLANTS PER PAR
     1ENT(RECIPS ABSENT OR POOLED)ARE)
      680 FORMAT (1H ,10F10.1)
      690 FORMAT (1H0,107H WHENEVER A DEPENDENCY(SINGULAR MATRIX) OCCURS THE
     1 GCA EFFECT IS SET TO ZERO. THE NUMBER OF DEPENDENCIES IS ,I3)
      700 FORMAT (1H0,28H THE MEAN(2GCA+XBAR)VALUE IS,E16.9)
      710 FORMAT (1H0,117H IF CORRECT DATA WERE ENTERED THE FOLLOWING WILL C
     1ORRESPOND TO TERMS FOUND BY INDEPENDENTLY DERIVED LEAST SQUARES AN
     2V)
      720 FORMAT (1H0,23H THE CORRECTION TERM IS,E16.9)
      730 FORMAT (1H0,31H THE SS TERM FOR GCA EFFECTS IS,E16.9)
      740 FORMAT (1H0,25H THE EXPERIMENTAL MEAN IS,E16.9)
      750 FORMAT (1H1,51HMATRIX INVERSE FOR DETERMINING CONFIDENCE INTERVALS
     1)
      760 FORMAT (1H1,64H THERE WERE RECIPROCAL CROSSES.THE ORIGINAL PLANTS
     1 /FAMILY WERE)
      770 FORMAT (1H1,71H THERE WERE RECIPROCAL CROSSES.THE DATA ENTERED--
     1 MEANS OR SUMS--WERE)
      780 FORMAT (1H1,39H THE ORIGINAL DATA ENTERED AS SUMS WERE)
      790 FORMAT (1H1,24H THE MEAN GCA VALUES ARE)
      800 FORMAT (1H ,5E20.9)
      810 FORMAT (1H1,26H THE (2GCA+XBAR)VALUES ARE)
      820 FORMAT (1H ,5E20.9)
      830 FORMAT (1H1,20X,1HI,5X,1HJ,7X,1HN,14X,4HXBAR,14X,3HSCA//)
      840 FORMAT (1H ,19X,I2,4X,I2,3E16.9)
C
        END
```

```
C     SUBROUTINE MPRINT
C
C     THIS SUBROUTINE WILL PRINTOUT SQUARE MATRICES UP TO 50X50
C     ALL ROWS AND COLUMNS ARE APPROPRIATELY LABELED.
C     DEFINITIONS OF SUBROUTINES ARGUMENTS
C     RAY(I,J) = MATRIX TO BE PRINTED
C     M=ORDER(MAXIMUM OF 50)
C     L=1 FOR 6E20.9 OUTPUT OR 2 FOR 10F12.2
C     TI IS A SIX-CHARACTER DESIGNATE OF THE MATRIX
C
      SUBROUTINE MPRINT (RAY,M,L,TI)
      DIMENSION RAY(50,50),JJ(50)
      IF(L-1)2,2,4
    2 L1=5
      GO TO 5
    4 L1 = 9
    5 J1 = 0
      J2 = 0
      JSEC = 0
      DO 8 I= 1,M
    8 JJ(I)=I
    9 J1 = J2+1
      J2 = J1+L1
      IF(J2-M)13,13,12
   12 J2=M
   13 JSEC = JSEC + 1
      IF(JSEC - 1) 18,18,19
   18 WRITE(6 ,17) TI,JSEC
   17 FORMAT(1H0,A6,9H SECTION ,I3/)
      GO TO 201
   19 WRITE(6 ,20) TI,JSEC
   20 FORMAT(1H1,A6,9H SECTION ,I3)
  201 IF(L-1)21,21,26
   21 WRITE(6 ,22) (JJ(I), I=J1,J2)
   22 FORMAT(6H0 ROW ,3X,I12,5I20)
      DO 23 I=1,M
   23 WRITE(6,24) I, (RAY(I,K), K=J1,J2)
   24 FORMAT(I6,4X,6E20.9)
      GO TO 31
   26 WRITE(6 ,27) (JJ(I),I=J1,J2)
   27 FORMAT(6H0 ROW ,3X,10I11)
      DO 29 I=1,M
   29 WRITE(6,30) I, (RAY(I,K),K=J1,J2)
   30 FORMAT(I6,4X,10F11.4)
   31 IF(J2-M)9,32,32
   32 RETURN
      END
```

```
C      SUBROUTINE MATINV MODIFIED TO HANDLE DEPENDENCIES
C
C      PROGRAMMED BY BURTON S. GARBOW, ARGONNE NATIONAL LABORATORY,
C      AND REPORTED IN IBM 704-709 SHARE LIBRARY AS AN F402.
C
C      THIS SUBROUTINE COMPUTES THE INVERSE AND DETERMINANT OF MATRIX A,
C      OF ORDER N, BY THE GAUSS-JORDAN METHOD. A-INVERSE REPLACES A,
C      AND THE DETERMINANT OF A IS PLACED IN DETERM. IF M = 1 THE
C      VECTOR BB CONTAINS THE CONSTANT VECTOR WHEN MATINV IS CALLED,
C      AND THIS IS REPLACED WITH THE SOLUTION VECTOR. IF M = 0, NO
C      SIMULTANEOUS EQUATION SOLUTIONS ARE CALLED FOR, AND BB IS NOT
C      PERTINENT. NN IS NOT TO EXCEED 50. A, NN, BB, M, AND DETERM IN
C      THE ARGUMENT LIST ARE DUMMY VARIABLES.
C      IORDER=1  ROWS OR COLUMNS NOT REORDERED
C             =0 REORDERED
C
C
       SUBROUTINE MATINV (A,NN,BB,M,DETERM,NDEP,IORDER)
C      SUBROUTINE MATINV DIMENSION
       DIMENSION IPIVOT(50),A(50,50),BB(50,1),INDEX(50,2),PIVOT(50)
       EQUIVALENCE (IROW,JROW), (ICOLUM,JCOLUM), (AMAX, TT, SWAP)
C      INITIALIZATION
       NDEP= 0
       EPS= 1.0E-06
    10 DETERM=1.0
    15 DO 20 J=1,NN
    20 IPIVOT(J)=0
    30 DO 555 I = 1, NN
C      SEARCH FOR PIVOT ELEMENT
C      FOLLOWING ALLOWS FOR NO REORDERING
       IF(IORDER.EQ.0) GO TO 40
       IROW=I
       ICOLUM=I
       GO TO 110
    40 AMAX=0.0
    45 DO 105 J=1,NN
    50 IF(IPIVOT(J)-1) 60, 105, 60
    60 DO 100 K=1,NN
    70 IF(IPIVOT(K)-1) 80, 100, 740
    80 IF(ABS(AMAX)-ABS(A(J,K))) 85, 100, 100
    85 IROW=J
    90 ICOLUM=K
    95 AMAX=A(J,K)
   100 CONTINUE
   105 CONTINUE
   110 IPIVOT(ICOLUM)=IPIVOT(ICOLUM)+1
C      INTERCHANGE ROWS TO PUT PIVOT ELEMENT ON DIAGONAL
   130 IF(IROW-ICOLUM) 140, 260, 140
   140 DETERM=-DETERM
   150 DO 200 L=1,NN
   160 SWAP=A(IROW,L)
   170 A(IROW,L)=A(ICOLUM,L)
   200 A(ICOLUM,L)=SWAP
   205 IF(M) 260, 260, 210
   210 DO 250 L=1,M
   220 SWAP=BB(IROW,L)
   230 BB(IROW,L)=BB(ICOLUM,L)
   250 BB(ICOLUM,L)=SWAP
   260 INDEX(I,1)=IROW
   270 INDEX(I,2)=ICOLUM
   310 PIVOT(I)=A(ICOLUM,ICOLUM)
       IF(ABS(PIVOT(I)).GT.EPS) GO TO 320
       IF(M.NE.1) GO TO 600
       DO 3700 L=1,M
```

```
 3700 BB(ICOLUM,L)=0.
      DETERM=0
      NDEP = NDEP +1
      GO TO 555
  320 DETERM=DETERM*PIVOT(I)
C     DIVIDE PIVOT ROW BY PIVOT ELEMENT
  330 A(ICOLUM,ICOLUM)=1.0
  340 DO 350 L=1,NN
  350 A(ICOLUM,L)=A(ICOLUM,L)/PIVOT(I)
  355 IF(M) 380, 380, 360
  360 DO 370 L=1,M
  370 BB(ICOLUM,L)=BB(ICOLUM,L)/PIVOT(I)
C     REDUCE NON-PIVOT ROWS
  380 DO 550 L1=1,NN
  390 IF(L1-ICOLUM) 400, 550, 400
  400 TT=A(L1,ICOLUM)
  420 A(L1,ICOLUM)=0.0
  430 DO 450 L=1,NN
  450 A(L1,L)=A(L1,L)-A(ICOLUM,L)*TT
  455 IF(M) 550, 550, 460
  460 DO 500 L=1,M
  500 BB(L1,L)=BB(L1,L)-BB(ICOLUM,L)*TT
  550 CONTINUE
  555 CONTINUE
C     INTERCHANGE COLUMNS
  600 DO 710 I=1,NN
  610 L=NN+1-I
  620 IF(INDEX(L,1)-INDEX(L,2)) 630, 710, 630
  630 JROW=INDEX(L,1)
  640 JCOLUM=INDEX(L,2)
  650 DO 705 K=1,NN
  660 SWAP=A(K,JROW)
  670 A(K,JROW)=A(K,JCOLUM)
  700 A(K,JCOLUM)=SWAP
  705 CONTINUE
  710 CONTINUE
  740 RETURN
      END
```

☆ U.S. GOVERNMENT PRINTING OFFICE: 1975—674—351

Snyder, E. B.

1975. Combining-ability determinations for incomplete mating designs. South. For. Exp. Stn., New Orleans, La. 12 p. (USDA For. Serv. Gen. Tech. Rep. SO-9)

It is shown how general combining ability values (GCA's) from cross-, open-, and self-pollinated progeny can be derived in a single analysis. Breeding values are employed to facilitate explaining genetic models of the expected family means and the derivation of the GCA's. A FORTRAN computer program also includes computation of specific combining ability values and several options.

Additional keywords: Diallel cross, reciprocal cross, specific combining ability, *panmixia*.

Snyder, E. B.

1975. Combining-ability determinations for incomplete mating designs. South. For. Exp. Stn., New Orleans, La. 12 p. (USDA For. Serv. Gen. Tech. Rep. SO-9)

It is shown how general combining ability values (GCA's) from cross-, open-, and self-pollinated progeny can be derived in a single analysis. Breeding values are employed to facilitate explaining genetic models of the expected family means and the derivation of the GCA's. A FORTRAN computer program also includes computation of specific combining ability values and several options.

Additional keywords: Diallel cross, reciprocal cross, specific combining ability, *panmixia*.